Result list for written exam in Mat1 spring syllabus 2021

rev. 180521

Initalisation

```
> restart:with(plots):with(LinearAlgebra):
> prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),
  Vector):
  vop:=proc(X) op(convert(X,list)) end proc:
  grad:=(X,Y)->convert(linalg[grad](X,Y),Vector[column]):
  div:=V->VectorCalculus[Divergence](V):
  rot:=proc(X) uses VectorCalculus;BasisFormat(false);Curl(X)
  end proc:
```

MapleTA Problems (in selected version)

A function f of two real variables is given by

```
> f:=(x,y)->x^2*y-8*x*y+y^2+y+1
```

$$f := (x, y) \mapsto x^2 \cdot y - 8 \cdot x \cdot y + y^2 + y + 1 \tag{2.1}$$

A parametrized curve is given by

```
> r:=u-><u,1-u^2>
```

$$r := u \mapsto \langle u, 1 - u^2 \rangle \tag{2.2}$$

We consider the composite function h(u) = f(r(u)).

Q1, a

Determine the two values of u for which h'(u) = -8

```
> f(vop(r(u))):
  simplify(%)
```

$$8 u^3 - 2 u^2 - 8 u + 3 \tag{2.1.1}$$

```
> diff(%,u)
```

$$24 u^2 - 4 u - 8 \tag{2.1.2}$$

```
> solve(%=-8,u)
```

$$0, \frac{1}{6} \tag{2.1.3}$$

u1 = 0 and u2 = 1/6.

Q1, b

Enter the coordinates of the pointt (x0,y0,z0) located on the graph surface of g vertically above the point r(u2).

```
> r(1/6)
```

(2.2.1)

$$\begin{bmatrix} \frac{1}{6} \\ \frac{35}{36} \end{bmatrix} \quad \text{(2.2.1)}$$

```
> f(vop(%))
```

$$\frac{89}{54} \quad \text{(2.2.2)}$$

x0 = 1/6
y0 = 35/36
z0 = 89/54

A function f of two variables is given by

```
> f:=(x,y)->x^2*y-4*x^2-8*x*y+y^2+32*x+8*y-44
```

$$f := (x, y) \mapsto x^2 \cdot y - 4 \cdot x^2 - 8 \cdot x \cdot y + y^2 + 32 \cdot x + 8 \cdot y - 44 \quad \text{(2.3)}$$

It is stated that the function f has just one stationary point (x0,y0),

Q2, a

Enter the coordinates of the stationary point

```
> lign1:=diff(f(x,y),x)
```

$$lign1 := 2\,x\,y - 8\,x - 8\,y + 32 \quad \text{(2.3.1)}$$

```
> lign2:=diff(f(x,y),y)
```

$$lign2 := x^2 - 8\,x + 2\,y + 8 \quad \text{(2.3.2)}$$

```
> solve({lign1,lign2},{x,y})
```

$$\{x = 4, y = 4\} \quad \text{(2.3.3)}$$

x0 = 4
y0 = 4

Q2, b

Determine the partial derivatives of second order for f in the point (x0,y0).

```
> D[1,1](f)(4,4);
  D[1,2](f)(4,4);
  D[2,1](f)(4,4);
  D[2,2](f)(4,4);
```

$$0$$
$$0$$
$$0$$
$$2 \quad \text{(2.4.1)}$$

Q2, c

We consider the bounded and closed set M bounded by the parabola with the equation y = -x^2+8x-11 and the line with the equation y =1.

Determine the maximum and the minimum value that f attains on M.

```
> plot({-x^2+8*x-11,1},x=0..8,view=0..5,scaling=constrained)
```



Where do they intersect?

```
> solve(-x^2+8*x-11=1)
```

$$2, 6 \tag{2.5.1}$$

The stationary point is an interior point.

```
> kandidat1=f(4,4)
```

$$kandidat1 = 4 \tag{2.5.2}$$

Now follows the boundary investigation:

The restriction of f to the parabola segment:

```
> f(x,-x^2+8*x-11)
```

$$x^2 \left(-x^2 + 8\,x - 11\right) - 12\,x^2 - 8\,x \left(-x^2 + 8\,x - 11\right) + \left(-x^2 + 8\,x - 11\right)^2 + 96\,x - 132 \tag{2.5.3}$$

```
> simplify(%)
```

$$-x^2 + 8\,x - 11 \tag{2.5.4}$$

```
> solve(diff(%,x))
```

$$4 \tag{2.5.5}$$

```
> kandidat2:=f(4,5)
```

$$kandidat2 := 5 \tag{2.5.6}$$

The restriction of f to the line y=1:

```
> f(x,1)
```

$$-3\,x^2 + 24\,x - 35 \tag{2.5.7}$$

```
> diff(%,x)
```

$$-6\,x + 24 \tag{2.5.8}$$

```
> kandidat3:=f(4,1)
```

$$kandidat3 := 13 \tag{2.5.9}$$

f in the intersection points:

```
> kandidat4=f(2,1);
  kandidat5=f(6,1)
```

$$kandidat4 = 1$$
$$kandidat5 = 1 \tag{2.5.10}$$

When comparing the five candidates:

The maximum value of f on M is 13 (attained in (4,1))

The minimum value of f on M is 1 (attained in the intersection points (2,1) and (6,1))

A set of points A in the (x,y)-plane is given by

$$A = \{ (x, y) \mid x \in [1, 2] \text{ and } y \in [0, \ln(x)] \}.$$



In addition a function of two real variables is given by f(x,y) = x-1

Q4

Determine the plane integral $\int_A f(x, y)\,d\mu$

p representation:

```
> r:=(u,v)-><u,0>+v*<0,ln(u)>:
> r(u,v)
```

$$\begin{bmatrix} u \\ v \ln(u) \end{bmatrix} \quad \textbf{(2.6.1)}$$

```
> M:=<diff(r(u,v),u)|diff(r(u,v),v)>
```

$$M := \begin{bmatrix} 1 & 0 \\ \frac{v}{u} & \ln(u) \end{bmatrix} \quad \textbf{(2.6.2)}$$

```
> Jacobi:=ln(u)
```

$$Jacobi := \ln(u) \quad \textbf{(2.6.3)}$$

```
> f:=(x,y)->x-1
```

$$f := (x, y) \mapsto x - 1 \quad \textbf{(2.6.4)}$$

```
> Int(f(vop(r(u,v)))*Jacobi,[u=1..2,v=0..1])
```

$$\int_0^1 \int_1^2 (u - 1) \ln(u)\, du\, dv \quad \textbf{(2.6.5)}$$

```
> value(%)
```

$$\frac{1}{4} \quad \textbf{(2.6.6)}$$

The plane integral = 1/4

Q5

In the (x,z)-plane in (x,y,z)-space we consider a profile curve K with the equation
$z = \cosh(x),\ x \in [0, 1]$.

Let F denote the surface of revolution that appears when K is rotated an angle of pi about the z-axis (positive direction of rotation as seen from the positive end of the z-axis).

Determine the mass of F with respect to the mass density function $f(x, y, z) = \frac{1}{z},\quad z \neq 0.$

```
> r:=(u,v)-><u*cos(v),u*sin(v),cosh(u)>:
> r(u,v)
```

$$\begin{bmatrix} u\cos(v) \\ u\sin(v) \\ \cosh(u) \end{bmatrix} \tag{2.7.1}$$

```
> kryds(diff(r(u,v),u),diff(r(u,v),v)):
  N:=simplify(%)
```

$$N := \begin{bmatrix} -\sinh(u)\, u\cos(v) \\ -\sinh(u)\, u\sin(v) \\ u \end{bmatrix} \tag{2.7.2}$$

```
> sqrt(prik(N,N));
  Jacobi:=simplify(%)assuming u::positive
```

$$\sqrt{\sinh(u)^2 u^2 \cos(v)^2 + \sinh(u)^2 u^2 \sin(v)^2 + u^2}$$

$$Jacobi := u\cosh(u) \tag{2.7.3}$$

```
> f:=(x,y,z)->1/z:
> Int(f(vop(r(u,v)))*Jacobi,[u=0..1,v=0..Pi])
```

$$\int_0^{\pi}\int_0^1 u \, du \, dv \tag{2.7.4}$$

```
> value(%)
```

$$\frac{\pi}{2} \tag{2.7.5}$$

The mass = Pi/2

Q7, a

Let k be a number greater than 0. there is in the (x,y)-plane in (x,y,z)-space given a filled rectangle B with vertices (0,-1,0), (0,1,0), (k,1,0) and (k,-1,0). In addition there is given the height function $h(x, y) = k - x \cdot y^2$

The figure below shoes the graph of the function h above the rectangle when k = 1.



Determine for k = 1 the volume of the solid region Ω which lies vertically between B and the graph for *h* .

```
> k:=1:
> r:=(u,v,w)-><u,v,w*(k-u*v^2)>:
> r(u,v,w)
```

$$\begin{bmatrix} u \\ v \\ w\left(-u\,v^2+1\right) \end{bmatrix} \quad \textbf{(2.8.1)}$$

```
> M:=<diff(r(u,v,w),u)|diff(r(u,v,w),v)|diff(r(u,v,w),w)>
```

$$M := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -w\,v^2 & -2\,w\,u\,v & -u\,v^2+1 \end{bmatrix} \quad \textbf{(2.8.2)}$$

```
> Determinant(M):
  Jacobi:=simplify(%)
```

$$Jacobi := -u\,v^2+1 \quad \textbf{(2.8.3)}$$

```
> Int(Jacobi,[u=0..k,v=-1..1,w=0..1])
```

$$\int_0^1\int_{-1}^1\int_0^1 \left(-u\,v^2+1\right)\,\mathrm{d}u\,\mathrm{d}v\,\mathrm{d}w \quad \textbf{(2.8.4)}$$

```
> value(%)
```

$$\frac{5}{3} \quad \textbf{(2.8.5)}$$

V = 5/3

Q7, b

Determine the value k must assume if the volume of Ω is equal to 25/3.

```
> k:='k':
> r:=(u,v,w)-><u,v,w*(k-u*v^2)>:
> r(u,v,w)
```

$$\begin{bmatrix} u \\ v \\ w\left(-u\,v^2+k\right) \end{bmatrix} \quad \textbf{(2.9.1)}$$

```
> M:=<diff(r(u,v,w),u)|diff(r(u,v,w),v)|diff(r(u,v,w),w)>
```

$$M := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -w\,v^2 & -2\,w\,u\,v & -u\,v^2 + k \end{bmatrix} \quad (2.9.2)$$

```
> Determinant(M):
  Jac:=simplify(%)
```

$$Jac := -u\,v^2 + k \quad (2.9.3)$$

```
> Int(Jac,[u=0..k,v=-1..1,w=0..1])
```

$$\int_0^1 \int_{-1}^1 \int_0^k \left(-u\,v^2 + k\right) \mathrm{d}u\,\mathrm{d}v\,\mathrm{d}w \quad (2.9.4)$$

```
> value(%)
```

$$\frac{5\,k^2}{3} \quad (2.9.5)$$

```
> solve(%=25/3)[1]
```

$$\sqrt{5} \quad (2.9.6)$$

k = sqrt(5)

Essay-assignment (in selected version, cf. pdf)

(the following is not an essay, but an extended list of answers)

In the (x,y,z)-space are given the vector fields

```
> V:=(x,y,z)-><z^2+2*y,3*y-1,x^2+x*z>:
> V(x,y,z)
```

$$\begin{bmatrix} z^2 + 2\,y \\ 3\,y - 1 \\ x^2 + x\,z \end{bmatrix} \quad (3.1)$$

and U(x,y,z) = Curl(V)(x,y,z).

```
> U:=(x,y,z)->rot(V)(x,y,z):
> U(x,y,z)
```

$$\begin{bmatrix} 0 \\ z - 2\,x \\ -2 \end{bmatrix} \quad (3.2)$$

```
> P:=<0,1,0>:
  Q:=<-1,0,0>:
  R:=<1,-1,0>:
  S:=<0,0,1>:
```

1)
The line from P to R, where u is in [0,1]:

```
> s1:=u-><u,-2*u+1,0>:
> s1(u)
```

$$\begin{bmatrix} u \\ -2\,u + 1 \\ 0 \end{bmatrix} \quad (3.3)$$

```
> integrand:=prik(V(vop(s1(u))),diff(s1(u),u))
```

$$integrand := -2 + 8\,u \quad (3.4)$$

The tangential curve integral:

```
> int(integrand,u=0..1)
```

$$2 \quad (3.5)$$

The line from P to Q, where u is in [0,1]:

```
> s2:=u->P+u*(Q-P):
  s2(u)
```

$$\begin{bmatrix} -u \\ 1 - u \\ 0 \end{bmatrix} \quad (3.6)$$

```
> integrand:=prik(V(vop(s2(u))),diff(s2(u),u))
```

$$integrand := -4 + 5\,u \quad (3.7)$$

```
> del1:=int(integrand,u=0..1)
```

$$del1 := -\frac{3}{2} \quad (3.8)$$

The line from Q to R, where u is in [0,1]:

```
> s3:=u->Q+u*(R-Q):
  s3(u)
```

$$\begin{bmatrix} -1 + 2\,u \\ -u \\ 0 \end{bmatrix} \quad (3.9)$$

```
> integrand:=prik(V(vop(s3(u))),diff(s3(u),u))
```

$$integrand := 1 - u \quad (3.10)$$

```
> del2:=int(integrand,u=0..1)
```

$$del2 := \frac{1}{2} \quad (3.11)$$

The tangential curve integral along the broken line

```
> del1+del2;
```

$$-1 \quad (3.12)$$

V cannot be a gradient field, as the tangential curve integrals from P to Q along two different paths are different.

2)

The triangle parameterized by straight lines from S to the points on the line segment PQ

where u and v are in [0,1]:

```
> r1:=(u,v)->S+v*(s2(u)-S)
```

$$r1 := (u, v) \mapsto S + v \cdot (s2(u) - S) \quad (3.13)$$

```
> r1(u,v)
```

$$\begin{bmatrix} -u\,v \\ v\,(1 - u) \\ 1 - v \end{bmatrix} \quad (3.14)$$

```
> plot3d(r1(u,v),u=0..1,v=0..1,axes=normal,labels=[x,y,z]):
> kryds(diff(r1(u,v),u),diff(r1(u,v),v)):
  N:=simplify(%)
```

$$N := \begin{bmatrix} v \\ -v \\ -v \end{bmatrix} \quad (3.15)$$

points downwards!

```
> prik(U(vop(r1(u,v))),N):
  integrand:=expand(%)
```

$$integrand := -2\,u\,v^2 + v^2 + v \quad (3.16)$$

The flux with respect to U:

```
> int(integrand,[u=0..1,v=0..1])
```

$$\frac{1}{2} \quad (3.17)$$

Eventually Stokes is used. If one uses the orientation PSQP, the circulation becomes as the preceding result, i.e. 1/2. With the opposite orientation, -1/2 is obtained.

3)

The triangle is parameterized by straight lines from Q to the line PR

where u and v are in [0,1]:

```
> r2:=(u,v)->Q+v*(s1(u)-Q)
```

$$r2 := (u, v) \mapsto Q + v\cdot(s1(u) - Q) \quad (3.18)$$

```
> r2(u,v);
```

$$\begin{bmatrix} -1 + v\,(u+1) \\ v\,(-2\,u + 1) \\ 0 \end{bmatrix} \quad (3.19)$$

```
> plot3d(r2(u,v),u=0..1,v=0..1,axes=normal,labels=[x,y,z]):
> kryds(diff(r2(u,v),u),diff(r2(u,v),v)):
  N:=simplify(%)
```

$$N := \begin{bmatrix} 0 \\ 0 \\ 3\,v \end{bmatrix} \quad (3.20)$$

pointing upwards as desired

```
> prik(V(vop(r2(u,v))),N):
  integrand:=expand(%)
```

$$integrand := 3\,u^2\,v^3 + 6\,u\,v^3 - 6\,u\,v^2 + 3\,v^3 - 6\,v^2 + 3\,v \quad (3.21)$$

The flux with respect to V:

```
> int(integrand,[u=0..1,v=0..1])
```

$$\frac{1}{4} \quad (3.22)$$

4)

The tetrahedron is parameterized by straight lines from S to the points in the triangle PQR

where u, v and w are in [0,1]:

```
> r3:=(u,v,w)->S+w*(r2(u,v)-S):
> r3(u,v,w)
```

(3.23)

$$\begin{bmatrix} w(-1+v(u+1)) \\ w\,v(-2u+1) \\ 1-w \end{bmatrix} \quad (3.23)$$

```
> M:=<diff~(r3(u,v,w),u)|diff~(r3(u,v,w),v)|diff~(r3(u,v,w),w)>;
```

$$M := \begin{bmatrix} w\,v & w(u+1) & -1+v(u+1) \\ -2\,w\,v & w(-2u+1) & v(-2u+1) \\ 0 & 0 & -1 \end{bmatrix} \quad (3.24)$$

```
> LinearAlgebra[Determinant](M);
  Jacobi:=-%;
```

$$-3\,w^2\,v$$

$$Jacobi := 3\,w^2\,v \quad (3.25)$$

```
> divV:=(x,y,z)->div(V)(x,y,z):
> divV(x,y,z)
```

$$3+x \quad (3.26)$$

```
> integrand:=divV(vop(r3(u,v,w)))*Jacobi;
```

$$integrand := 3\,(3+w(-1+v(u+1)))\,w^2\,v \quad (3.27)$$

Using Gauss, the flux is now found through the surface:

```
> int(integrand,[u=0..1,v=0..1,w=0..1])
```

$$\frac{3}{2} \quad (3.28)$$

5)

```
> lign1:=diff(x(t),t)=0;
  lign2:=diff(y(t),t)=z(t)-2*x(t);
  lign3:=diff(z(t),t)=-2;
```

$$lign1 := \frac{d}{dt}\,x(t)=0$$

$$lign2 := \frac{d}{dt}\,y(t)=z(t)-2\,x(t)$$

$$lign3 := \frac{d}{dt}\,z(t)=-2 \quad (3.29)$$

```
> lsn:=dsolve({lign1,lign2,lign3,x(0)=r1(u,v)[1],y(0)=r1(u,v)
  [2],z(0)=r1(u,v)[3]},{x(t),y(t),z(t)});
```

$$lsn := \{x(t)=-u\,v,\ y(t)=-t^2+2\,t\,u\,v+t\,(1-v)-u\,v+v,\ z(t)=-2\,t+1-v\} \quad (3.30)$$

```
> r:=unapply(<rhs(lsn[1]),rhs(lsn[2]),rhs(lsn[3])>,(u,v,t)):
  'r(u,v,t)'=r(u,v,t);
```

$$r(u,v,t)=\begin{bmatrix} -u\,v \\ -t^2+2\,t\,u\,v+t\,(1-v)-u\,v+v \\ -2\,t+1-v \end{bmatrix} \quad (3.31)$$

```
> r(u,v,0),r(u,v,1)
```

$$\begin{bmatrix} -u\,v \\ -u\,v+v \\ 1-v \end{bmatrix}, \begin{bmatrix} -u\,v \\ u\,v \\ -1-v \end{bmatrix} \quad (3.32)$$

```
> plot3d({r(u,v,0),r(u,v,1)},u=0..1,v=0..1,scaling=constrained)
```




```
> N:=kryds(diff(r(u,v,1),u),diff(r(u,v,1),v))
```

$$N := \begin{bmatrix} -v \\ -v \\ 0 \end{bmatrix} \tag{3.33}$$

```
> sqrt(prik(N,N))assuming v>0:
  Jacobi:=%
```

$$Jacobi := \sqrt{2}\, v \tag{3.34}$$

```
> Area=int(Jacobi,[u=0..1,v=0..1])
```

$$Area = \frac{\sqrt{2}}{2} \tag{3.35}$$